NAG Library Routine Document

C09CCF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

C09CCF computes the one-dimensional multi-level discrete wavelet transform (DWT). The initialization routine C09AAF must be called first to set up the DWT options.

2 Specification

```
SUBROUTINE C09CCF(N, X, LENC, C, NWL, DWTLEV, ICOMM, IFAIL)
INTEGER            N, LENC, NWL, DWTLEV(NWL+1), ICOMM(100), IFAIL
double precision   X(N), C(LENC)
```

3 Description

C09CCF computes the multi-level DWT of one-dimensional data. For a given wavelet and end extension method, C09CCF will compute a multi-level transform of a data array, x_i, for $i = 1, 2, \ldots, n$, using a specified number, n_l, of levels. The number of levels specified, n_l, must be no more than the value returned in NWL by the initialization routine C09AAF for the given problem. The transform is returned as a set of coefficients for the different levels (packed into a single array) and a representation of the multi-level structure.

The notation used here assigns level 0 to the input dataset, x, with level 1 being the first set of coefficients computed, with the detail coefficients, d_1, being stored while the approximation coefficients, a_1, are used as the input to a repeat of the wavelet transform. This process is continued until, at level n_l, both the detail coefficients, d_{n_l}, and the approximation coefficients, a_{n_l} are retained. The output array, C, stores these sets of coefficients in reverse order, starting with a_{n_l} followed by $d_{n_l}, d_{n_l - 1}, \ldots, d_1$.

4 References

None.

5 Parameters

1: N – INTEGER *Input*

On entry: the number of elements, n, in the data array x.

Constraint: This must be the same as the value N passed to the initialization routine C09AAF.

2: X(N) – ***double precision*** array *Input*

On entry: X contains the one-dimensional input dataset x_i, for $i = 1, 2, \ldots, n$.

3: LENC – INTEGER *Input*

On entry: the dimension of the array C as declared in the (sub)program from which C09CCF is called. C must be large enough to contain the number, n_c, of wavelet coefficients. The maximum value of n_c is returned in NWC by the call to the initialization routine C09AAF and corresponds to the DWT being continued for the maximum number of levels possible for the given data set. When the number of levels, n_l, is chosen to be less than the maximum, then n_c is correspondingly smaller and LENC can be reduced by noting that the number of coefficients at each level is given by $\mathrm{ceiling}(n/2)$ for MODE = 'P' in C09AAF and $\mathrm{floor}\big((n + n_f - 1)/2\big)$ for MODE = 'H', 'W', 'Z', where n is the number of the input data at that level and n_f is the filter length NF provided by the call to C09AAF. At the final level the storage is doubled to contain the set of approximation coefficients.

Constraint: LENC $\geq n_c$, where n_c is the number of approximation and detail coefficients that correspond to a transform with NWL levels.

4: C(LENC) – ***double precision*** array *Output*

On exit: the coefficients of a multi-level wavelet transform of the dataset.

If $n(a_{n_l})$ and $n(d_i)$, for $i = n_l, n_l - 1, \ldots, 1$, are the numbers of approximation and detail coefficients returned in DWTLEV where the subscript identifies the level and $n(a_{n_l})$ is stored in DWTLEV(1), then setting $k_1 = n(a_{n_l})$ and $k_{j+1} = n(a_{n_l}) + \sum_{i=1}^{j} n(d_{n_l-i+1})$, for $j = 1, 2, \ldots, n_l$, the coefficients are stored in C as follows:

C$(1 : k_1)$
Contains the level n_l approximation coefficients, a_{n_l}.

C$(k_1 + 1 : k_2)$
Contains the level n_l detail coefficients d_{n_l}.

C$(k_j + 1 : k_{j+1})$
Contains the level $n_l - j + 1$ detail coefficients, for $j = 2, \ldots, n_l$.

5: NWL – INTEGER *Input*

On entry: the number of levels in the multi-level resolution to be performed.

Constraint: $1 \leq \text{NWL} \leq n_l$, where n_l is the value returned in NWL (the maximum number of levels) by the call to the initialization routine C09AAF.

6: DWTLEV$(\text{NWL} + 1)$ – INTEGER array *Output*

On exit: the number of transform coefficients at each level. DWTLEV(1) and DWTLEV(2) contain the number of approximation, $n(a_{n_l})$, and detail coefficients, $n(d_{n_l})$, for the final level of resolution (these are equal); DWTLEV(i) contains the number of detail coefficients, $n(d_{n_l-i+2})$, for the $(n_l - i + 2)$th level, for $i = 3, \ldots, \text{NWL} + 1$.

7: ICOMM(100) – INTEGER array *Communication Array*

On entry: contains details of the discrete wavelet transform and the problem dimension as setup in the call to the initialization routine C09AAF.

On exit: contains additional information on the computed transform.

8: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N is inconsistent with the value passed to the initialization routine C09AAF.

IFAIL $= 3$

On entry, LENC $< n_c$, where n_c is the value returned in NWC by the call to the initialization routine C09AAF.

IFAIL $= 5$

On entry, NWL < 1,
or NWL $> n_l$, where n_l is the value returned in NWL by the call to the initialization routine C09AAF.

IFAIL $= 7$

On entry, the initialization routine C09AAF has not been called first or it has been called with WTRANS = 'S', or the communication array ICOMM has become corrupted.

7 Accuracy

The accuracy of the wavelet transform depends only on the floating point operations used in the convolution and downsampling and should thus be close to machine precision.

8 Further Comments

The wavelet coefficients at each level can be extracted from the output array C using the information contained in DWTLEV on exit (see the descriptions of C and DWTLEV in Section 5). For example, given an input data set, x, denoising can be carried out by applying a thresholding operation to the detail coefficients at every level. The elements $\mathrm{C}(k_1 + 1 : k_{n_l+1})$, as described in Section 5, contain the detail coefficients, $\hat{d}_{ij}$, for $i = n_l, n_l{-}1, \ldots, 1$ and $j = 1, 2, \ldots, n(d_i)$, where $\hat{d}_{ij} = d_{ij} + \sigma\epsilon_{ij}$ and $\sigma\epsilon_{ij}$ is the transformed noise term. If some threshold parameter α is chosen, a simple hard thresholding rule can be applied as

$$\bar{d}_{ij} = \begin{cases} 0, & \text{if } \left|\hat{d}_{ij}\right| \leq \alpha \\ \hat{d}_{ij}, & \text{if } \left|\hat{d}_{ij}\right| > \alpha, \end{cases}$$

taking $\bar{d}_{ij}$ to be an approximation to the required detail coefficient without noise, d_{ij}. The resulting coefficients can then be used as input to C09CDF in order to reconstruct the denoised signal. Note that the number of levels, n_l, stored in NWL and the numbers of coefficients contained in DWTLEV, must remain unchanged.

See the references given in the introduction to this chapter for a more complete account of wavelet denoising and other applications.

9 Example

This example performs a multi-level resolution of a dataset using the Daubechies wavelet (see WAVNAM = 'DB4' in C09AAF) using zero end extensions, the number of levels of resolution, the number of coefficients in each level and the coefficients themselves are reused. The original dataset is then reconstructed using C09CDF.

9.1 Program Text

```
*     C09CCF Example Program Text
*     Mark 22 Release. NAG Copyright 2007.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX, NWCMAX, LENC, NWLMAX
      PARAMETER        (NMAX=850,NWCMAX=850,LENC=NWCMAX,NWLMAX=20)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, N, NF, NNZ, NWC, NWL, NY
      CHARACTER*10     MODE, WAVNAM, WTRANS
*     .. Local Arrays ..
      DOUBLE PRECISION C(NWCMAX), X(NMAX), Y(NMAX)
      INTEGER          DWTLEV(NWLMAX), ICOMM(100)
*     .. External Subroutines ..
      EXTERNAL         C09AAF, C09CCF, C09CDF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'C09CCF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
*     Read problem parameters
      READ (NIN,*) N
      READ (NIN,*) WAVNAM, MODE
*
      IF (N.LE.NMAX .AND. N.GE.2) THEN
         WRITE (NOUT,99999) WAVNAM, MODE, N
*
*        Read data array and write it out
*
         READ (NIN,*) (X(I),I=1,N)
*
         WRITE (NOUT,*) ' Input Data      X :'
         WRITE (NOUT,99998) (X(I),I=1,N)
*
*        Query wavelet filter dimensions
*        For Multi-Resolution Analysis, decomposition, WTRANS = 'M'
*
         IFAIL = 1
         WTRANS = 'Multilevel'
         CALL C09AAF(WAVNAM,WTRANS,MODE,N,NWL,NF,NWC,ICOMM,IFAIL)
*
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99997) IFAIL
            GO TO 40
         END IF
*
*        Perform Discrete Wavelet transform
*
         IFAIL = 1
         CALL C09CCF(N,X,LENC,C,NWL,DWTLEV,ICOMM,IFAIL)
*
         IF (IFAIL.EQ.0) THEN
            WRITE (NOUT,99996) NWL
            WRITE (NOUT,99995)
            WRITE (NOUT,99994) (DWTLEV(I),I=1,NWL+1)
            NNZ = 0
            DO 20 I = 1, NWL + 1
               NNZ = NNZ + DWTLEV(I)
   20       CONTINUE
            WRITE (NOUT,99993)
            WRITE (NOUT,99998) (C(I),I=1,NNZ)
*
*           Reconstruct original data
*
            NY = N
            IFAIL = 1
            CALL C09CDF(NWL,LENC,C,NY,Y,ICOMM,IFAIL)
*
            IF (IFAIL.NE.0) THEN
               WRITE (NOUT,99992) IFAIL
```

```
            ELSE
               WRITE (NOUT,99991)
               WRITE (NOUT,99998) (Y(I),I=1,NY)
            END IF
         ELSE
            WRITE (NOUT,99990) IFAIL
         END IF
*
   40 END IF
*
99999 FORMAT (1X,' MLDWT :: Wavelet : ',A10,', End mode : ',A10,' N = ',
     +       I10)
99998 FORMAT (8(F8.4,1X),:)
99997 FORMAT (1X,' ** C09AAF returned with IFAIL = ',I5)
99996 FORMAT (1X,' Number of Levels : ',I10)
99995 FORMAT (1X,' Number of coefficients in each level : ')
99994 FORMAT (8(I8,1X),:)
99993 FORMAT (1X,' Wavelet coefficients C : ')
99992 FORMAT (1X,' ** C09CDF returned with IFAIL = ',I5)
99991 FORMAT (1X,' Reconstruction              Y : ')
99990 FORMAT (1X,' ** C09CCF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

```
C09CCF Example Program Data
64                                 : N
DB4  Zero                          : WAVNAM, MODE
6.5271 6.512  6.5016 6.5237 6.4625
6.3496 6.4025 6.4035 6.4407 6.4746
6.5095 6.6551 6.61   6.5969 6.6083
6.652  6.7113 6.7227 6.7196 6.7649
6.7794 6.8037 6.8308 6.7712 6.7067
6.769  6.7068 6.7024 6.6463 6.6098
6.59   6.596  6.5457 6.547  6.5797
6.5895 6.6275 6.6795 6.6598 6.6925
6.6873 6.7223 6.7205 6.6843 6.703
6.647  6.6008 6.6061 6.6097 6.6485
6.6394 6.6571 6.6357 6.6224 6.6073
6.6075 6.6379 6.6294 6.5906 6.6258
6.6369 6.6515 6.6826 6.7042        : X(1:N)
```

9.3 Program Results

```
 C09CCF Example Program Results
  MLDWT :: Wavelet : DB4       , End mode : Zero       N =         64
  Input Data      X :
  6.5271   6.5120   6.5016   6.5237   6.4625   6.3496   6.4025   6.4035
  6.4407   6.4746   6.5095   6.6551   6.6100   6.5969   6.6083   6.6520
  6.7113   6.7227   6.7196   6.7649   6.7794   6.8037   6.8308   6.7712
  6.7067   6.7690   6.7068   6.7024   6.6463   6.6098   6.5900   6.5960
  6.5457   6.5470   6.5797   6.5895   6.6275   6.6795   6.6598   6.6925
  6.6873   6.7223   6.7205   6.6843   6.7030   6.6470   6.6008   6.6061
  6.6097   6.6485   6.6394   6.6571   6.6357   6.6224   6.6073   6.6075
  6.6379   6.6294   6.5906   6.6258   6.6369   6.6515   6.6826   6.7042
  Number of Levels :          6
  Number of coefficients in each level :
       7        7        8       10       14       21       35
  Wavelet coefficients C :
  0.0000  -0.0227  -0.3446   2.7574 -10.1970  44.8800  15.9443   0.0010
 -0.4881 -10.2673  11.3258  -1.7469   2.0785  -0.7334  -0.0054  -0.1402
 -5.8980  -1.1527   5.5613   2.1352   0.3203  -0.4004   0.0010   0.5229
  0.5055  -2.7274  -0.0911  -0.2806  -0.3669   2.9467  -0.3799  -0.1552
  0.0218   0.0922   5.4626  -2.1620   0.5196  -0.0287  -0.0199   0.0920
 -0.0134  -0.1298  -5.5168   2.3105  -0.5383  -0.0155   0.3057   0.6186
 -1.5542   0.2682   0.1566   0.0030  -0.0152  -0.0589   0.0126   0.0063
  0.0171  -0.0268   0.0077  -0.0189   0.0207   0.0104  -0.3207  -0.6062
  1.6288  -0.2414  -0.0671   3.1657  -1.1462   0.2785   0.0523  -0.0030
 -0.0270  -0.0442   0.0090   0.0171  -0.0230  -0.0015   0.0213  -0.0402
 -0.0263  -0.0099   0.0021  -0.0250   0.0210  -0.0028  -0.0298  -0.0095
```

```
  0.0034   0.0281  -0.0188  -0.0002  -0.0173  -0.0076  -0.0014   0.0184
 -0.0318   0.0048   0.0047  -3.2555   1.1710  -0.2913
  Reconstruction           Y :
  6.5271   6.5120   6.5016   6.5237   6.4625   6.3496   6.4025   6.4035
  6.4407   6.4746   6.5095   6.6551   6.6100   6.5969   6.6083   6.6520
  6.7113   6.7227   6.7196   6.7649   6.7794   6.8037   6.8308   6.7712
  6.7067   6.7690   6.7068   6.7024   6.6463   6.6098   6.5900   6.5960
  6.5457   6.5470   6.5797   6.5895   6.6275   6.6795   6.6598   6.6925
  6.6873   6.7223   6.7205   6.6843   6.7030   6.6470   6.6008   6.6061
  6.6097   6.6485   6.6394   6.6571   6.6357   6.6224   6.6073   6.6075
  6.6379   6.6294   6.5906   6.6258   6.6369   6.6515   6.6826   6.7042
```